
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test cases 8.1.3.2.2 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-06_D19wk38’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

2. Table of Contents

21. Overview

2. Table of Contents
2
3. Verification Test Summary
2
4. Corrections required for NR5GC test case 8.1.3.2.2 using wk38
2
5. Execution Log Files
9
5.1 HiSilicon Balong 5000
9
References
9

3. Verification Test Summary

Test Case:
8.1.3.2.2
Test Group:
\NR5GC_IRAT\8_1_3\
ATS Version:
iwd-TTCN3-B2019-06_D19wk38
System Simulator used:
Starpoint SP9500 Test system

UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC test case 8.1.3.2.2 using wk38
Change 4.1

	Template name
	cmr_EUTRA_NR_SysInfo

	Reason for change
	Add new template which can receive both NR sysInfo and Eutra sysinfo since f_EUTRA_NR_WaitForCoOrd_SysInfo is used in NR and EUTRA port.

	Summary of change
	Add new template

	TTCN module
	\Common\Common4G5G\EUTRA_NR_CoordMsg.ttcn

	MCC160 Comment
	


Before change

      template (present) EUTRA_NR_Coordination_MSG cmr_EUTRA_NR_SysInfo :=

  {

    SysInfo := ?

  };

Change 4.2

	Function name
	f_EUTRA_NR_WaitForCoOrd_SysInfo

	Reason for change
	Since f_EUTRA_NR_WaitForCoOrd_SysInfo is used in NR and EUTRA port to receive sysinfo, so only NR sysinfo template is not correct when receiving EUTRA cell info.

	Summary of change
	Use new template in change 4.1 to receive EUTRA and NR cell sysinfo

	TTCN module
	\Common\Common4G5G\EUTRA_NR_CoordMsg.ttcn

	MCC160 Comment
	


Before change

     function f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA_NR_CO_ORD_PORT p_Port) return EUTRA_NR_CoOrd_SysInfo_Type

  {

    var EUTRA_NR_Coordination_MSG v_ReceivedMsg;

    alt {

      [] p_Port.receive(cmr_EUTRA_NR_SysInfoNR) -> value v_ReceivedMsg {}

      [] a_EUTRA_NR_Default(p_Port) {}

    }

    return v_ReceivedMsg.SysInfo;

  }

After change

  function f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA_NR_CO_ORD_PORT p_Port) return EUTRA_NR_CoOrd_SysInfo_Type

  {

    var EUTRA_NR_Coordination_MSG v_ReceivedMsg;

    alt {

      [] p_Port.receive(cmr_EUTRA_NR_SysInfo) -> value v_ReceivedMsg {}

      [] a_EUTRA_NR_Default(p_Port) {}

    }

    return v_ReceivedMsg.SysInfo;

  }
Change 4.3

	Function name
	f_TC_8_1_3_2_2_NR5GC_EUTRA

	Reason for change
	Since the max powe of EUTRA cell is -73, so initial Max reference Power should be set to -73.

	Summary of change
	Set max reference Power to -73

	TTCN module
	\NR5GC_IRAT\8_1_3\RRC_Measurement_NR5GC_EUTRA.ttcn

	MCC160 Comment
	


Before change

   function f_TC_8_1_3_2_2_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B2 / Measurement of E-UTRA cells

    var integer v_EUTRA_CellPower_T0_FR1 := -73;

    f_EUTRA38_IRAT_Init(c31);

    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T0_FR1, v_EUTRA_CellPower_T0_FR1);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_3_2_2_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }
After change

    function f_TC_8_1_3_2_2_NR5GC_EUTRA() runs on EUTRA_5GS_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B2 / Measurement of E-UTRA cells

    var integer v_EUTRA_CellPower_T0_FR1 := -73;

    f_EUTRA38_IRAT_Init(c31);

f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -73);
    //Create and configure cell

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_NR_WaitForCoOrd_Trigger(NR);

    f_EUTRA38_SetCellPower(eutra_Cell1, v_EUTRA_CellPower_T0_FR1, v_EUTRA_CellPower_T0_FR1);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_3_2_2_NR5GC_EUTRA_TestBody ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

    f_EUTRA_ReleaseAllCells();

  }

Change 4.4

	Function name
	f_TC_8_1_3_2_2_NR5GC()

	Reason for change
	1. According to 38508-1 subclause 4.4.3.1.2, SIB5 is broadcast in NR-6.

2. Send postamble to tell EUTRA postamble is finished

	Summary of change
	1. Use f_NR5GC_IRAT_Init to broadcast SIB5
2. Add trigger after Postamble

	TTCN module
	\NR5GC_IRAT\8_1_3\RRC_Measurement_NR5GC_NR.ttcn

	MCC160 Comment
	


Before change

     function f_TC_8_1_3_2_2_NR5GC() runs on NR5GC_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B2 / Measurement of E-UTRA cells

    f_NR5GC_Init(NR_6);
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -75, -75); //Maximum power level for FR2 FFS

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(true);

    f_TC_8_1_3_2_2_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  }// End of f_TC_8_1_3_2_2_NR5GC()
After change

    function f_TC_8_1_3_2_2_NR5GC() runs on NR5GC_PTC

  { // Measurement configuration control and reporting / Inter-RAT measurements / Event B2 / Measurement of E-UTRA cells

     f_NR5GC_IRAT_Init(NR_6);  // System information combination NR-6 as defined in TS 38.508-1 clause 4.4.3.1.2 is used in NR Cell
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -75, -75); //Maximum power level for FR2 FFS

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(true);

    f_TC_8_1_3_2_2_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

    // Now tell EUTRA Postamble finished

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger("Postamble"));
  }// End of f_TC_8_1_3_2_2_NR5GC()

Change 4.5

	Function name
	f_TC_8_1_3_2_2_TestBody

	Reason for change
	1. According to R5-198920, Threshold EUTRA for B2 is 58.
2. MeasGapUE should be used, not gapFR1 since according to 38.133 subclause 9.1.2, per-FR gap is support only when UE supports two independent measurement gap.
3. According to 38.331 subclause5.5.5, the measurement quantities for serving cell may have the available SINR. 

	Summary of change
	1. Change 73 to 58.
2. Use per-UE gap.
3. Usingthe default measQuantityResults which is based on pc_ss_SINR_Meas.

	TTCN module
	\NR5GC_IRAT\Common\EUTRA38_NR5GCProcedures.ttcn

	MCC160 Comment
	


Before change

     function f_TC_8_1_3_2_2_TestBody() runs on NR_BASE_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var NR_AbsoluteCellPower_Type v_CellPower_T0_FR1 := -75;

    var NR_AbsoluteCellPower_Type v_CellPower_T1_FR1 := -85;

    var PhysCellId v_PhysCellId;

    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;

    var EUTRA_PhysCellId v_EUTRA_CellId;

    var integer v_MeasurementBandwidth_int;

    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;

    var Hysteresis v_Hysteresis := 4;

    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;

    var RSRP_Range v_B2_Threshold1 := 76; //@sic R5-195346 sic@

    var RSRP_RangeEUTRA v_B2_Threshold2EUTRA := 73; //@sic R5-195346 sic@

    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;

    var template (value) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    //Power levels of NR cell 1: Table 8.1.3.2.2.3.2-1, Table 8.1.3.2.2.3.2-2

    /*         |    FR1     |    FR2

     * -------------------------------

     *   T0    |    -75     |    FFS

     * -------------------------------

     *   T1    |    -85     |    FFS

     * -------------------------------

     */

    v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo (EUTRA);

    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;

    v_EUTRA_CellId := bit2int (v_CoOrd_SysInfo.Eutra[0].CellId);

    v_MeasurementBandwidth_int := v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth;

    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_MeasurementBandwidth_int);

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T0_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T1_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B2

    //The UE transmits an RRCReconfigrationComplete message

    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};

    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};

    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB2_EUTRA(v_B2_Threshold1, v_B2_Threshold2EUTRA, v_Hysteresis, v_ReportAmount, false, false))};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_MeasGapConfig_FR1, cs_38508_QuantityConfig_EUTRA);

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig);

    //@siclog "Steps  3" siclog@

    //Check: Does the UE transmit a MeasurementReport message within the next 10s?

    f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event B2 with the measured RSRP value for NR Cell 1?

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId, tsc_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp)}), cr_NR_MeasQuantityResults_RsrpRsrq);//Removed
  }//End of f_TC_8_1_3_2_2_TestBody()
After change

    function f_TC_8_1_3_2_2_TestBody() runs on NR_BASE_PTC

  {

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var NR_AbsoluteCellPower_Type v_CellPower_T0_FR1 := -75;

    var NR_AbsoluteCellPower_Type v_CellPower_T1_FR1 := -85;

    var PhysCellId v_PhysCellId;

    var ARFCN_ValueEUTRA v_ARFCN_ValueEUTRA;

    var EUTRA_PhysCellId v_EUTRA_CellId;

    var integer v_MeasurementBandwidth_int;

    var EUTRA_AllowedMeasBandwidth v_MeasurementBandwidth;

    var Hysteresis v_Hysteresis := 4;

    var EventTriggerConfigInterRAT.reportAmount v_ReportAmount := r1;

    var RSRP_Range v_B2_Threshold1 := 76; //@sic R5-195346 sic@

    var RSRP_RangeEUTRA v_B2_Threshold2EUTRA := 58; //@sic R5-195346 sic@

    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList_NR;

    var template (value) MeasObjectParamsList_Type_EUTRA v_MeasuredObjectList_EUTRA;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (value) MeasIdToAddModList v_MeasIdConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    //Power levels of NR cell 1: Table 8.1.3.2.2.3.2-1, Table 8.1.3.2.2.3.2-2

    /*         |    FR1     |    FR2

     * -------------------------------

     *   T0    |    -75     |    FFS

     * -------------------------------

     *   T1    |    -85     |    FFS

     * -------------------------------

     */

    v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA);

    v_ARFCN_ValueEUTRA := v_CoOrd_SysInfo.Eutra[0].Arfcn;

    v_EUTRA_CellId := bit2int (v_CoOrd_SysInfo.Eutra[0].CellId);

    v_MeasurementBandwidth_int := v_CoOrd_SysInfo.Eutra[0].DL_Bandwidth;

    v_MeasurementBandwidth := f_NR5GC_IRAT_ConvertIntegerToEUTRA_AllowedMeasBandwidth(v_MeasurementBandwidth_int);

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T0_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_T1_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)};

    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to setup inter-RAT measurement and reporting for event B2

    //The UE transmits an RRCReconfigrationComplete message

    v_MeasuredObjectList_NR := {cs_MeasObjectId1(nr_Cell1)};

    v_MeasuredObjectList_EUTRA := {cs_MeasObjectId2_EUTRA(v_EUTRA_CellId, v_ARFCN_ValueEUTRA)};

    v_ReportConfigList := {cs_NR_ReportConfig_reportConfigInterRAT(tsc_NR_IdReportConfigId1, cs_NR_ReportConfigInterRAT_Trigger_eventB2_EUTRA(v_B2_Threshold1, v_B2_Threshold2EUTRA, v_Hysteresis, v_ReportAmount, false, false))};

    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR_EUTRA(v_ARFCN_ValueEUTRA, v_MeasurementBandwidth, -, v_MeasuredObjectList_NR, v_MeasuredObjectList_EUTRA, v_ReportConfigList, v_MeasIdConfigList, cs_38508_MeasGapConfig_UE(cs_38508_GapConfig), cs_38508_QuantityConfig_EUTRA);

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig);

    //@siclog "Steps  3" siclog@

    //Check: Does the UE transmit a MeasurementReport message within the next 10s?

    f_NR_Check_NoMeasurementReport(10.0, nr_Cell1, cr_NR_MeasurementReport_Any);

    //@siclog "Step  4" siclog@

    //SS re-adjusts the cell-specific reference signal level according to row "T1" in Table 8.1.3.2.1.3.2-1/2.

    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 5" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event B2 with the measured RSRP value for NR Cell 1?

    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellId, tsc_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_EUTRA({cr_NR_MeasResultEUTRA_CellResults(v_EUTRA_CellId, cr_NR_MeasQuantityResultsEUTRA_Rsrp)}));

  }//End of f_TC_8_1_3_2_2_TestBody()  

5. Execution Log Files

5.1 HiSilicon Balong 5000

The HiSilicon Balong 5000 UE passed this test case on Starpoint SP9500 Test System on NR band n41-FDD3.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
\ TC_8_1_3_2_2_NR5GC_PASS.spm
PIXCIT settings used:

 \ NR5GC_IRAT_Testsuite.sppar

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

